# Sushi rating and ranking:

## Combining Plackett-Luce & ordinal logit in Stan

**Bob Carpenter**

Center for Computational Mathematics

Flatiron Institute

September 2024

# Prelude: The Bradley-Terry model

- **Bradley and Terry** (1952) modeled **consumer preference ranking**

  e.g., prefer soap 17 to soap 23

- $R \in \mathbb{N}$ raters and $I \in \mathbb{N}$ items

- **Data**: $y_{i,j} = 1$ if item $i$ preferred to item $j$ and $0$ otherwise

- **Parameters**: each item $i \in 1{:}I$ gets a score $\alpha_i \in \mathbb{R}$.
  - set $\mathrm{sum}(\alpha) = 0$ (built-in soon!) or $\alpha_1 = 0$ to **identify** likelihood

- **Model**: $\mathrm{Pr}[\text{rater prefers } i \text{ to } j] = \mathrm{logit}^{-1}(\alpha_i - \alpha_j) = \dfrac{\exp(\alpha_i)}{\exp(\alpha_i) + \exp(\alpha_j)}$

- **Sampling distribution**: $y_{i,j} \sim \mathrm{bernoulli}\big(\mathrm{logit}^{-1}(\alpha_i - \alpha_j)\big)$

# Partial panel data

- Often use **multiple raters** on subsample of item set.
  - some pairs may be rated more than once
  - some pairs may never be rated
- $N$ rankings $y_n \in \{0, 1\}$ of item $a_n \in 1{:}I$ vs. $b_n \in 1{:}I$.
- **Sampling distribution**: $y_n \sim \mathrm{bernoulli}\left(\mathrm{logit}^{-1}(\alpha_{a[n]} - \alpha_{b[n]})\right)$.
  - $a[n]$ used for $a_n$ in subscripts for readability

# Diving in: rating and ranking data

- **Toshiro Kamishima** (2003) collected data on sushi preferences
  - $R = 5000$ number of raters (**Japanese people**)
  - $I = 100$ number of items (**types of sushi**, e.g., fatty tuna, shrimp, $\ldots$)
  - $K = 10$ number of ratings per rater (**incomplete panel**)
  - $x_r \in (1{:}I)^K$: pieces rated by rater $r$ (no duplicates)
    e.g., $x_1 = 58, 4, 3, 44, 87, 60, 67, 1, 12, 74$
  - $y_{r,k} \in 1{:}K$ **rank** of item $k$ ($x_{r,k}$) by rater $r$ ($y_r$ a permutation)
    e.g., $y_1 = 3, 2, 8, 4, 5, 10, 1, 7, 6, 9$
  - $z_{r,k} \in 1{:}5$ **rating** of item $k$ ($x_{r,k}$) by rater $r$ on 1–5 scale
    e.g., $z_1 = 2, 2, 5, 3, 3, 5, 1, 4, 4, 5$
- Same human feedback format used to “align” **ChatGPT** responses

# The Luce Axiom

- Given a set of items, **choice among any subset** must have **proportional probabilities**.
- For example,
  - suppose three ice cream flavors, vanilla, chocolate, strawberry.
  - If choosing among all three, probabilities are $0.7, 0.2$, and $0.1$, then
  - chocolate vs. strawberry must be $\dfrac{0.2}{0.2+0.1} = \frac{2}{3}$ and $\dfrac{0.1}{0.2+0.1} = \frac{1}{3}$, and
  - vanilla vs. strawberry must be be $\dfrac{0.7}{0.7+0.1} = \frac{7}{8}$ and $\dfrac{0.1}{0.7+0.1} = \frac{1}{8}$.

# Plackett-Luce model of ranking

- Plackett-Luce is an **iterative Bradley-Terry model**
- Each item still gets a score $\alpha_i$ (same **non-identifiability**)
- **Next choice model**: $\Pr[\text{rank } i_1 \text{ ahead of } i_2, \ldots, l_N] = \dfrac{\exp(\alpha_1)}{\exp(\alpha_1) + \cdots + \exp(\alpha_N)}$
- **Ranking model**:

$$
\begin{aligned}
\Pr[\text{rank } i_1 > i_2 > \cdots > i_N] \quad = \quad & \Pr[\text{rank } i_1 > i_2, \ldots, i_N] \\
& \times \Pr[\text{rank } i_2 > i_3, \ldots, i_N] \\
& \vdots \\
& \times \Pr[\text{rank } i_{N-1} > i_N]
\end{aligned}
$$

# Stan model for Plackett-Luce

```
data {
  int<lower=1> I;  // # items
  int<lower=1> K;  // # items ranked per rater
  int<lower=1> R;  // # raters
  array[R, K] int<lower=1, upper=I> y;  // rankings
} parameters {
  simplex[I] beta;  // complete selection probabilities
}
model {
  for (r in 1:R) y[r] ~ plackett_luce(beta);
}
```

- Simplex $\beta = \mathrm{softmax}(\alpha) = \dfrac{\exp(\alpha)}{\mathrm{sum}(\exp(\alpha))}$

# Plackett-Luce density in Stan

```
real plackett_luce_lpmf(array[] int y, vector beta) {
  vector[size(y)] beta_y = beta[y];
  return sum(log(beta_y ./ cumulative_sum(beta_y)));
}
```

- Directly codes Luce axiom with $\alpha = \log \beta$
- The Plackett-Luce density is a product of fractions, for which
  - `beta[y]` indexes numerator values $\exp(\alpha)$, and
  - cumulative sums computes all denominators.
- Return sum of logs to put product on log scale.

# Fit of Plackett-Luce

- **Easy to fit**—usually a sign model matches data well
- Following is from **50 warmup** and **50 sampling** iterations in one chain

| | Mean | StdDev | 5% | 95% | N_Eff | R_hat |
|---|---|---|---|---|---|---|
| beta[1] | 0.018 | 0.0006 | 0.017 | 0.019 | 34 | 1.00 |
| beta[2] | 0.016 | 0.0004 | 0.015 | 0.017 | 71 | 1.00 |
| beta[3] | 0.025 | 0.0008 | 0.023 | 0.026 | 47 | 1.00 |
| beta[4] | 0.014 | 0.0004 | 0.014 | 0.015 | 57 | 0.98 |
| .... | | | | | | |

- Effective sample size of 30+ is plenty for inference.

# The results for best sushi are . . .

```
rank  score   type (description)
   1  0.0360  chu_toro  (mildly_fatty tuna)
   2  0.0342  toro  (fatty tuna)
   3  0.0246  maguro  (tuna)
   4  0.0230  negi_toro  (fatty flesh of tuna minced to a paste and
   5  0.0227  amaebi  (AMA shrimp)
   6  0.0226  kurumaebi  (prawn)
   7  0.0218  negi_toro_maki  (roll style of no.37)
   8  0.0210  samon  (salmon)
   9  0.0205  tarabagani  (king crab)
  10  0.0204  tai  (sea bream)
```

- **(Semi)-fatty tuna by a mile—central** $90\%$ **intervals** are roughly $\pm 0.002$

# And the results for worst . . .

```
rank  score   type (description)
  91  0.0039  nasu  (egg plant)
  92  0.0039  ankimo  (angler liver)
  93  0.0039  kyabia  (caviar)
  94  0.0035  mamakari  (Japanese scaled sardine)
  95  0.0034  karasumi  (dried mullet roe)
  96  0.0033  himo_kyu_maki  (part of clam & cucumber roll)
  97  0.0033  kue  (kind of cabrilla)
  98  0.0032  hoya  (ascidian)
  99  0.0031  okura  (gumbo)
 100  0.0015  namako  (sea cucumber)
```

- uncertainty of $\pm 0.002$ makes **exact ranking uncertain**

# Ordinal logistic rating model

- **Data**: **rating** $z_n \in 1{:}K$ and **piece rated** $\text{item}_n \in 1{:}I$
- **Parameters**: quality $\eta_i$ for each piece (constrained to sum to zero); four **cutpoints** $c_1 < c_2 < c_3 < c_4$
- **Sampling distribution**: $z_n \sim \text{OrderedLogistic}(\eta_{\text{item}[n]}, c)$, where

$$\text{OrderedLogistic}(k \mid \eta, c) = \begin{cases} 1 - \text{logit}^{-1}(\eta - c_1) & \text{if } k = 1 \\ \text{logit}^{-1}(\eta - c_{k-1}) - \text{logit}^{-1}(\eta - c_k) & \text{if } 1 < k < K \\ \text{logit}^{-1}(\eta - c_{K-1}) - 0 & \text{if } k = K \end{cases}$$

# Visualizing 4-outcome ordinal logit model




- **Facets**: $\eta \in \{-2, 0, 4\}$; **Blue lines**: cutpoints $c = \begin{bmatrix} -1.2 & 0.3 & 2.9 \end{bmatrix}$
- **Black line**: standard logistic density centered at $\eta$
- **Area under density** between vertical lines is **ordinal probability**

# Ordinal logit in Stan

```
data {
  int<lower=1> I;                           // # items
  int<lower=1> K;                           // # items per rater
  int<lower=0> R;                           // # raters
  int<lower=0> J;                           // maximum rating
  array[R, K] int<lower=1, upper=I> u;      // items rated
  array[R, K] int<lower=1, upper=J> z;      // ordinal ratings
}
parameters {
  ordered[K - 1] c;                         // cutpoints
  sum_to_zero_vector[I] eta;                // item quality
}
model {
  eta ~ normal(0, 3);     c ~ normal(0, 3);   // weakly informative
  z ~ ordered_logistic(eta[u], c);            // likelihood built-in
}
```

# Ordinal logit fit

- Fits well with high ESS/iteration and all $\widehat{R} \approx 1$

| | Mean | StdDev | 5% | 95% | N_Eff | R_hat |
|---|---|---|---|---|---|---|
| c[1] | -2.300 | 0.018 | -2.30 | -2.200 | 58.0 | 0.99 |
| c[2] | -1.400 | 0.016 | -1.40 | -1.400 | 49.0 | 0.98 |
| c[3] | -0.110 | 0.013 | -0.13 | -0.085 | 68.0 | 0.99 |
| c[4] | 0.920 | 0.016 | 0.89 | 0.950 | 39.0 | 0.99 |
| ... | ... | ... | ... | ... | ... | ... |
| eta[100] | -0.750 | 0.160 | -1.00 | -0.470 | 85.0 | 1.00 |

# Top-ten rating comparison

```
 ORDINAL-LOGIT                 PLACKETT-LUCE
 1  1.68  toro                 1  0.036  chu_toro
 2  1.68  chu_toro             2  0.034  toro
 3  1.50  negi_toro            3  0.025  maguro
 4  1.35  maguro               4  0.023  negi_toro
 5  1.27  amaebi               5  0.023  amaebi
 6  1.27  negi_toro_maki       6  0.023  kurumaebi
 7  1.27  tarabagani           7  0.022  negi_toro_maki
 8  1.04  ebi                  8  0.021  samon
 9  1.02  kurumaebi            9  0.021  tarabagani
10  1.02  samon               10  0.020  tai
```

- tai is 16th in ordinall logit; ebi is 12th in Plackett-Luce
- coefficients less separated in Plackett-Luce model

# How do coefficients compare?




- rating vs. ranking (left) and rating vs. log ranking (right) coefficients.

# Joint model

- Use a single simplex parameter $\beta$ for ranking.
- Let $\eta = \alpha_0 + \alpha_1 \cdot \beta$.
- Use both data sets and likelihoods.

# Top-ten rating comparison

| ORDINAL-LOGIT | | | PLACKETT-LUCE | | | COMBO | | |
|---|---|---|---|---|---|---|---|---|
| 1 | 1.68 | toro | 1 | 0.036 | chu_toro | 1 | 0.0354 | chu_toro |
| 2 | 1.68 | chu_toro | 2 | 0.034 | toro | 2 | 0.0337 | toro |
| 3 | 1.50 | negi_toro | 3 | 0.025 | maguro | 3 | 0.0250 | negi_toro |
| 4 | 1.35 | maguro | 4 | 0.023 | negi_toro | 4 | 0.0248 | maguro |
| 5 | 1.27 | amaebi | 5 | 0.023 | amaebi | 5 | 0.0230 | amaebi |
| 6 | 1.27 | negi_toro_maki | 6 | 0.023 | kurumaebi | 6 | 0.0224 | negi_toro_m |
| 7 | 1.27 | tarabagani | 7 | 0.022 | negi_toro_m | 7 | 0.0212 | kurumaebi |
| 8 | 1.04 | ebi | 8 | 0.021 | samon | 8 | 0.0212 | tarabagani |
| 9 | 1.02 | kurumaebi | 9 | 0.021 | tarabagani | 9 | 0.0203 | samon |
| 10 | 1.02 | samon | 10 | 0.020 | tai | 10 | 0.0189 | tai |

- tai is 16th in ordinall logit; ebi is 12th in Plackett-Luce
- coefficients less separated in Plackett-Luce model

# What's missing?

- Could group ratings by rater and model **rater accuracy/bias**
  - not much data to do that
  - crowdsourcing problems are what got me into Bayesian stats
- The **raters have covariates** we could use

  including geography and tastes vary regionally
- The items fall into groups which could be used

  e.g., fish vs. shellfish vs. vegetable, tuna vs. other, etc.
- Cross-validation evaluation (e.g., LOO) and posterior predictive checks.
- The weakly informative **priors seem fine**.